

# 33

## ADAPTATION: PERSISTENCE IN A CHANGING ENVIRONMENT

*In collaboration with Mary Puterbaugh*

### Objectives

- Consider how recombination and natural selection can lead to new phenotypes.
- Develop a spreadsheet model of allele and genotype frequencies at three loci.
- Examine how the abruptness of an environmental change affects the ability of a population to adapt to that change.
- Consider how genetic factors (recombination, genetic diversity, and number of genes) influence the likelihood of extinction in a finite population experiencing selective pressure.

*Suggested Preliminary Exercise: Hardy-Weinberg Equilibrium*

### INTRODUCTION

We hear a lot these days about global warming. Global climate change is not a new phenomenon—over its history, the earth has been warmer than it is today, and also much, much colder. But one of the concerns biologists have about the current warming trend is that, because it is occurring so rapidly, many populations will not be able to respond to the changes.

For many organisms even a small increase in environmental temperature can spell the difference between life and death. Estuarine marine organisms, for example, may have to adapt quickly to rising sea levels in order to persist over time. Species that cannot adapt quickly will go extinct, while species that are able to adapt will persist. What factors govern whether a population persists through a period of environmental change? Population size is obviously one answer. But we also should consider whether enough *heritable genetic variation* is present to allow the population to respond to selective pressures. Such variation arises either through mutation or recombination. This exercise will illustrate the process of recombination, an important force for evolution as we understand it.

**Recombination** is the process by which a sperm or an egg randomly receives one allele from a pair of alleles possessed by each parent. Suppose your mother has the genotype  $A_1A_1B_1B_1C_1C_1$  for the  $A$ ,  $B$ , and  $C$  loci, and your father has the genotype  $A_2A_2B_2B_2C_2C_2$ . You must have the genotype  $A_1A_2B_1B_2C_1C_2$ , because each of your parents produced only one type of allele at those loci, and you inherited one allele from each parent for each locus. In your case, however your **gametes** (eggs or sperm) randomly receive either an  $A_1$  or  $A_2$  allele, a  $B_1$  or  $B_2$  allele, and a

$C_1$  or  $C_2$  allele during meiosis. Your gametes thus have the potential to carry any one of the following *nine* genotypes:  $A_1B_1C_1$ ,  $A_1B_2C_1$ ,  $A_1B_2C_2$ ,  $A_2B_1C_2$ ,  $A_2B_1C_1$ ,  $A_1B_1C_2$ ,  $A_2B_2C_1$ , or  $A_2B_2C_2$ . Your mother could produce only  $A_1B_1C_1$  eggs, but the alleles you inherited from your father *recombined* with hers to create genotypes (yours!) that weren't present in the previous generation.

Recombination has a strong influence on the genotypes of offspring, especially for traits that are controlled by multiple genes. For example, beak size in birds is a heritable trait, and many different genes probably act together to determine beak size for an individual bird. When many genes affect the expression of a single trait, it is called a **polygenic trait**. Many traits are polygenic. In the simplest case, each locus makes a contribution to the expressed trait. For example, three different loci ( $A$ ,  $B$ , and  $C$ ) might contribute to beak size. If an individual inherits an  $A_1$ ,  $B_1$ , or  $C_1$  allele from its parents, it "inherits" a 1-mm contribution to beak size. If it inherits an  $A_2$ ,  $B_2$ , or  $C_2$  allele from its parents, it "inherits" a 2-mm contribution to beak size. Thus,  $A_1A_1B_1B_1C_1C_1$  individuals have the smallest beaks (6 mm), while  $A_2A_2B_2B_2C_2C_2$  individuals have the largest beaks (12 mm). Individuals that are heterozygous at either gene have intermediate-sized beaks (e.g.,  $A_1A_2B_1B_2C_1C_2$  genotypes have 9-mm beaks). The loci, then, act **additively** to determine the phenotype. Because several loci contribute to beak size, the population will tend to exhibit **continuous variation** in beak size, with beaks ranging from 6 mm to 12 mm.

The environment may play a large role in determining which genotype combinations are "best suited" in terms of survival and reproduction. For example, large beak size in one of Darwin's finches (*Geospiza fortis*) may be favored in drought years, but small beak sizes may be favored in wet years (Grant and Grant 1993). In other words, certain genotype combinations are favored under drought conditions, while other combinations are favored under wet conditions. Imagine for a moment that the frequencies of the alleles  $A_2$ ,  $B_2$ , and  $C_2$  (the alleles that produce larger beaks) are initially low in a given population. This means that  $A_2A_2$  individuals will be rare, as will  $B_2B_2$  and  $C_2C_2$  individuals. The probability that random mating and recombination will produce an individual with the genotype  $A_2A_2B_2B_2C_2C_2$  may be so small that this genotype may never occur in the population. If natural selection favors larger beaks, however, the frequencies of the  $A_2$ ,  $B_2$ , and  $C_2$  alleles in the population will increase, and recombination may occasionally produce individuals with the  $A_2A_2B_2B_2C_2C_2$  genotype.

Experiments with corn and fruit flies have demonstrated dramatic changes in phenotype that are probably the result of selection and recombination. In a famous experiment, Clayton and Robertson (1957) started out with a population of fruit flies and counted the bristles on the abdomen of each fly. They found that the number of bristles varied from 30 to 50. Over many generations, Clayton and Robertson consistently took the flies with the highest number of bristles and mated them. After 35 generations, all of the flies had between 60 and 110 bristles—phenotypes that didn't even occur in the original population!

Perhaps some novel mutation arose that increased bristle number, but it is more likely that changes in the frequencies of existing alleles led to the changes in bristle number: If bristle number is polygenic—controlled by several different genes—and if alleles that produce higher bristle numbers are rare, then the probability may be very small that recombination will produce an individual with more than 50 bristles. But by selecting *against* individuals with low bristle numbers (or *for* individuals with high bristle numbers), Clayton and Robertson increased the frequencies of the alleles that produce high bristle numbers, and thus increased the probability that recombination would result in individuals with more than 50 bristles. After 35 generations, the frequencies of alleles that result in high bristle numbers were high enough that recombination occasionally produced individuals with 110 bristles. On the other hand, the frequencies of alleles that produce low bristle numbers decreased, making it very unlikely that recombination could produce an individual with fewer than 60 bristles.

## PROCEDURES

From an evolutionary perspective, key questions include “How much genetic variation is needed for a population to persist through a period of rapid environmental change?” and “How does variation in environmental conditions affect the ability of a population to respond?” In this exercise, you’ll set up a spreadsheet model to answer these questions. We will consider a single trait (beak size, which determines drought resistance) and the allele frequencies at different loci that influence beak size. To begin, the initial allele frequencies will be determined by you, the modeler. The population at the beginning of the first year will consist of 500 adults with beak sizes determined by the allele frequencies you input. This population will then experience the environmental conditions for that year, again determined by the modeler. Certain individuals will survive to reproduce, while others will not. Those that survive will go on to reproduce at the end of the year. Since beak size is a heritable trait, the “new” population will have beak sizes that reflect the genetic composition of the survivors. You will follow the population for 5 years, during which you can alter environmental conditions (dry, mild, and wet) and alter the phenotypes that survive. If in any given year no individuals survive, the population has gone extinct. If you are pinched for time, you may model just 3 generations.

The goal of the exercise is to explore how much genetic variation is needed for a population to adapt to environmental change, and to explore how variation in environmental conditions affects the genetic diversity of populations. As always, save your work frequently to disk.

### INSTRUCTIONS

#### A. Set up the model population.

1. Open a new spreadsheet and set up column headings as shown in Figure 1.

2. In cells C4–E4, enter 11, 8, and 0 respectively as shown.

3. Enter the environmental conditions shown in cells B11–B15.

### ANNOTATION

	A	B	C	D	E	F	G	H
1	<i>Adaptation</i>							
2								
3	<b>Environmental condition:</b>		<b>Dry</b>	<b>Mild</b>	<b>Wet</b>			
4	<b>Select phenotypes above:</b>		11	8	0			
5								
6								
7			<b>Allele frequencies of surviving parents</b>					
8	<b>Phenotype contribution =&gt;</b>		<b>1</b>	<b>2</b>	<b>1</b>	<b>2</b>	<b>1</b>	<b>2</b>
9	<b>Year</b>	<b>Condition</b>	<b>A1</b>	<b>A2</b>	<b>B1</b>	<b>B2</b>	<b>C1</b>	<b>C2</b>
10	<b>Initial</b>		0.8	0.2	0.8	0.2	0.8	0.2
11	<b>1</b>	<b>Wet</b>						
12	<b>2</b>	<b>Mild</b>						
13	<b>3</b>	<b>Mild</b>						
14	<b>4</b>	<b>Dry</b>						
15	<b>5</b>	<b>Mild</b>						

Figure 1

We’ll consider three types of environmental conditions: dry, mild, and wet. Each condition favors different beak-length phenotypes. If a year is wet, individuals with a beak size of greater than 0 will survive. If there is a severe drought (dry conditions), only individuals with beak sizes greater than 11 will survive.

For now, year 1 will be a wet year, years 2, 3, and 5 will be mild, and year 4 will be a dry year. You will be able to manipulate these environmental conditions later in the exercise.

4. Enter the initial allele frequencies of the population shown in cells C10–H10.

5. Enter the phenotypic contributions of each allele as shown in cells C8–H8.

6. Save your work.

**B. Track the population through year 1.**

1. Set up the new column headings shown in Figure 2, but extend and repeat your column headings to 5 years.

2. In cells A20–A519, establish a population of 500 individuals.

3. In cell B20, enter a formula to generate a genotype for individual 1, and copy the formula down to obtain genotypes for the remaining individuals in the population.

4. In cell C20, enter a formula to generate phenotypes for each individual. Copy your formula down

We'll track the allele frequencies of three loci (*A*, *B*, and *C*) over a 5-year period. At each locus, there are just two alleles; their frequencies are *p* and *q*. In year 1, we'll start with allele frequencies of roughly  $A_1 = B_1 = C_1 = 0.8$ . Because only two alleles are present at each locus, the frequencies of the  $A_2$ ,  $B_2$ , and  $C_2$  alleles must be  $(1 - p)$ , which is 0.2.

These contributions ultimately determine what an individual's phenotype will be. For example, the number 1 entered in cell C8 designates that individuals with the  $A_1$  allele inherit a 1-mm contribution to beak size. The number 2 in cell D8 specifies that individuals with the  $A_2$  allele inherit a 2-mm contribution to beak size. With the phenotypic contributions given, the genotype  $A_1A_2B_1B_2C_2C_2$  has a phenotype of  $1 + 2 + 1 + 2 + 2 + 2$ , or 10 mm.

Repeat the column headings **Genotype**, **Phenotype**, **Survive?** and **Phenotype** for years 2 – 5 in columns F through U.

	A	B	C	D	E
18	Year 1				
19	Individual	Genotype	Phenotype	Survive?	Phenotype

**Figure 2**

Enter the number 1 in cell A20.  
In cell A21, enter **=A20+1**. Copy this formula down to cell A519.

Now we will assign genotypes to individuals at the beginning of year 1. These genotypes depend on the allele frequencies of breeders from the previous year, listed as "initial" frequencies. Only some of these genotypes will actually survive to breed at the end of the year. You might review the formulas used in the Hardy-Weinberg exercise.

Enter the formula **=IF(RAND()<C\$10,"A1","A2")&IF(RAND()<C\$10,"A1","A2")&IF(RAND()<E\$10,"B1","B2")&IF(RAND()<E\$10,"B1","B2")&IF(RAND()<G\$10,"C1","C2")&IF(RAND()<G\$10,"C1","C2")** in cell B20. Copy this formula down the column.

Each individual will have two alleles at each of the three loci (*A*, *B*, and *C*); the three loci are joined with the & symbol. (In the above rendition, the formula for each allele is on a separate line; your formula will be entered as a unit, with no spaces around the ampersands). Let's go over the formula for the *A* locus: Have the spreadsheet generate a random number. If this number is less than the allele frequency for the  $A_1$  allele given in cell C10, return an  $A_1$  allele; if the random number is greater than the allele frequency of the  $A_1$  allele given in cell C10, return an  $A_2$ . Use the analogous procedure to generate the second allele at the *A* locus, and then to obtain the *B* and *C* alleles.

Enter the formula **=LOOKUP(MID(B20,1,2),C\$9:\$H\$9,C\$8:\$H\$8)+LOOKUP(MID(B20,3,2),C\$9:\$H\$9,C\$8:\$H\$8)+LOOKUP(MID(B20,5,2),C\$9:\$H\$9,C\$8:\$H\$8)+LOOKUP(MID(B20,7,2),C\$9:\$H\$9,C\$8:\$H\$8)+**

to obtain phenotypes for the remaining individuals in the population.

**LOOKUP(MID(B20,9,2),\$C\$9:\$H\$9,\$C\$8:\$H\$8)+  
LOOKUP(MID(B20,11,2),\$C\$9:\$H\$9,\$C\$8:\$H\$8)** in cell C20 (there should be no spaces when you enter the formula). Copy this formula down the column.

We used two functions to generate phenotypes: the **LOOKUP** and **MID** functions. The **MID** function returns a specific number of characters from a text string, starting at the position you specify. It has the syntax **MID(text,start\_num,num\_chars)**, where **text** is the text string containing the characters you want to extract, **start\_num** is the position of the first character you want to extract in text, and **num\_chars** is the number of characters you want to extract. The first character in text has **start\_num 1**, and so on.

For example, **=MID(B20,1,2)** tells the spreadsheet to examine the genotype in cell B20, start with the first character in the genotype, and return two characters. If your genotype in cell B20 is **A1A1B1B1C1C1**, the **MID** function will return the portion of the genotype that is bolded. Similarly, the formula **=MID(B20,5,2)** will examine the genotype in cell B20, start with the fifth character in the genotype, and return two characters (the program will return "B1").

The **LOOKUP** formula returns a value either from a one-row or one-column range or from an array. The **LOOKUP** function has two syntax forms: vector and array. We will use the vector form, which looks in a one-row or one-column range (the vector) for a value and returns a value from the same position in a second one-row or one-column range. It has the syntax **LOOKUP(lookup\_value,lookup\_vector,result\_vector)**, where **lookup\_value** is a value the function searches for in the first vector, **lookup\_vector** is a range that contains only one row or one column, and **result\_vector** is the value that the spreadsheet returns from the same position in a row or column that is adjacent to the lookup vector. For example, **=LOOKUP("A1",C9:H9,C8:H8)** finds the value  $A_1$  in the vector C9–H9 and returns the phenotype contribution associated with that allele.

We have combined **LOOKUP** and **MID** formulae to generate a phenotype. For example, **=LOOKUP(MID(B20,1,2),\$C\$9:\$H\$9,\$C\$8:\$H\$8)** uses the **MID** formula to determine the first allele in the *A* locus (either  $A_1$  or  $A_2$ ), finds this value in cells C9–H9, and returns the associated phenotype contribution listed in cells C8–H8. You can add several of these kinds of formulae together to generate a final phenotype. It produces a very long formula that looks intimidating at first, but is really quite simple once you work through it.

5. In cell D20, enter a formula to determine whether individual 1 survived the conditions associated with year 1. Copy the formula down for the remaining individuals in the population.

Enter the formula **=IF(C20>LOOKUP(\$B\$11,\$C\$3:\$E\$3,\$C\$4:\$E\$4),B20,"")** in cell D20. Copy this formula down the column.

We want to know whether an individual survives to reproduce, given the environmental condition for year 1 (cell B11) and the beak size required to survive the environment for year 1 (listed in cells C4–E4). The formula simply tells the spreadsheet to look up year 1's condition in cell B11, locate that condition in cells C3–E3, and return the minimum phenotype required for survival listed in cells C4–E4. **IF** the individual has a phenotype greater than necessary for survival, return the individual's *genotype*; otherwise, return a blank cell (indicated by the two sets of quotation marks). Year 1 is a wet condition, and hence all genotypes will survive.

6. In cell E20, enter a formula that returns the individual's phenotype if it survived.

Enter the formula **=IF(D20="","",C20)** in cell E20. Copy this formula down for the remaining 499 individuals in the population.

7. Set up new headings as shown in Figure 3.

	I	J
8	Number	Mean
9	surviving	phenotype
10		
11		
12		
13		
14		
15		

Figure 3

8. In cell I11, enter a formula to count the number of survivors in year 1. These individuals will produce offspring for the next generation.

9. In cell J11, use the **AVERAGE** function to calculate the mean phenotype of the survivors.

10. Enter formulae in cells C11–H11 to compute allele frequencies of the surviving adults.

Enter the formula `=COUNTIF(D20:D519,"A*")` in cell I11.

The **COUNTIF** formula counts the number of cells within a range that meet the given criteria. The formula above tells the spreadsheet to examine cells D19–D518 and to count any cell that begins with an *A*. The \* following the *A* is a wild card, indicating that it doesn't matter what text follows the *A*. Since only surviving individuals have a genotype listed, the formula will count only those individuals that survived.

Enter the formula `=AVERAGE(E20:E519)` in cell J11.

We entered the following formulae:

- Cell C11 `=(2*COUNTIF($D$20:$D$519,"A1A1*")+COUNTIF($D$20:$D$519,"A1A2*")+COUNTIF($D$20:$D$519,"A2A1*"))/(2*$I$11)`
- Cell D11 `=1-C11`
- Cell E11 `=(2*COUNTIF($D$20:$D$519,"*B1B1*")+COUNTIF($D$20:$D$519,"*B1B2*")+COUNTIF($D$20:$D$519,"*B2B1*"))/(2*$I$11)`
- Cell F11 `=1-E11`
- Cell G11 `=(2*COUNTIF($D$20:$D$519,"*C1C1*")+COUNTIF($D$20:$D$519,"*C1C2*")+COUNTIF($D$20:$D$519,"*C2C1*"))/(2*$I$11)`
- Cell H11 `=1-G11`

You have entered similar formulae in your Hardy-Weinberg exercise. Remember the trick of using the \* wild card character. For example, when we used the **COUNTIF** formula to count the number of *A1A1\** individuals, it counted all individuals with the *A1A1* genotype, regardless of their genotypes at the *B* or *C* locus. The same principle applies to the *B* (*\*B1B1\**) and *C* (*\*C1C1*) genotypes.

Since these individuals survived to breed, they will determine the genotypes of individuals at the beginning of year 2.

Your spreadsheet should now look something like Figure 4. Your numbers will be a bit different in Row 11, and that's fine.

11. Save your work.

	A	B	C	D	E	F	G	H
7			Allele frequencies of surviving parents					
8	Phenotype contribution =>		1	2	1	2	1	2
9	Year	Condition	A1	A2	B1	B2	C1	C2
10	Initial		0.8	0.2	0.8	0.2	0.8	0.2
11	1	Wet	0.80	0.20	0.78	0.22	0.81	0.19

Figure 4

### C. Track the population for year 2.

1. In cells F20–F519, enter a formula to generate a genotype for each individual (offspring), given the allele frequencies listed in cells C11–H11.

2. Select cell C20, and copy it to cell G20.

3. Enter a formula in cell H20 to determine if individual 1 survives to breed in year 2.

4. Select cell E20, and copy it to cell I20.

5. Enter a formula in cell I12 to count the number of survivors in year 2.

6. Enter a formula in cell J12 to determine the average phenotype of survivors in year 2.

7. Enter formulae in cells C12–H12 to compute the allele frequencies of survivors for year 2.

The headings in Figure 5 should already be in place. You can simply repeat the step you completed for year 1 to complete column F.

Enter the formula `=IF(RAND()<C$11,"A1","A2")&IF(RAND()<C$11,"A1","A2")&IF(RAND()<E$11,"B1","B2")&IF(RAND()<E$11,"B1","B2")&IF(RAND()<G$11,"C1","C2")&IF(RAND()<G$11,"C1","C2")` in cell F19. Copy this formula down to row F519.

	F	G	H	I
18	Year 2			
19	Genotype	Phenotype	Survive?	Phenotype

Figure 5

This will determine the phenotypes of the 500 individuals that are present in the population at the beginning of year 2.

Refer back to the formula used in year 1. We entered the formula `=IF(G20>LOOKUP($B$12,$C$3:$E$3,$C$4:$E$4),F20,"")`.

This formula looks up the conditions associated with year 2 and returns the phenotype of individuals whose beak sizes are large enough to survive the environmental conditions for year 2.

The formula in cell E20 returns the phenotype of individuals that survive to breed.

Enter the formula `=COUNTIF(H20:H519,"A*")` in cell I12.

Enter the formula `=AVERAGE(I20:I518)` in cell J12.

As you did for year 1, compute the allele frequencies for the population that survives to breed in year 2. These frequencies will be used to assign genotypes to individuals (offspring) in year 3.

- Cell C12  $= (2 * \text{COUNTIF}(\$H\$20:\$H\$519, "A1A1*") + \text{COUNTIF}(\$H\$20:\$H\$519, "A1A2*") + \text{COUNTIF}(\$H\$20:\$H\$519, "A2A1*")) / (2 * \$I\$12)$
- Cell D12  $= 1 - C12$
- Cell E12  $= (2 * \text{COUNTIF}(\$H\$20:\$H\$519, "*B1B1*") + \text{COUNTIF}(\$H\$20:\$H\$519, "*B1B2*") + \text{COUNTIF}(\$H\$20:\$H\$519, "*B2B1*")) / (2 * \$I\$12)$
- Cell F12  $= 1 - E12$
- Cell G12  $= (2 * \text{COUNTIF}(\$H\$20:\$H\$519, "*C1C1*") + \text{COUNTIF}(\$H\$20:\$H\$519, "*C1C2*") + \text{COUNTIF}(\$H\$20:\$H\$519, "*C2C1*")) / (2 * \$I\$12)$
- Cell H12  $= 1 - G12$

8. Save your work.

9. Repeat steps 1–8 to obtain results for each of years 3–5 in cells J20–U519.

**D. Create graphs.**

1. Graph the frequencies of each allele over time.

Note that when you press F9, the calculate key, the spreadsheet generates new genotypes, and hence a new set of survivors and frequencies.

Use the line graph option and label your axes fully. Your graph should resemble Figure 6.




Figure 6

2. Graph the numbers of survivors over the 5-year period.

Your graph should resemble Figure 7.




Figure 7

## QUESTIONS

1. Hit the F9 key 20 times, keep track of the values in cells I11–I15, and count how many times the population goes extinct. (Under these conditions it will probably never go extinct.) In what percentage of the 20 trials did the population go extinct at any time during the 5-year period? This is the *extinction rate* for the situation in which year 1 is wet, years 2, 3, and 5 have a mild drought, and year 4 is dry (drought conditions). Change cell B12 (year 2) to DRY instead of MILD. Again hit the F9 key 20 times. In what percentage of the 20 trials did the population go extinct? This is the extinction rate for the situation in which the change in precipitation occurred more abruptly. Relate the extinction rate to the genetic variation and phenotypic variation in the population.
2. How do starting initial allele frequencies affect how the population adapts to abrupt changes in environmental conditions?
3. What if initial frequency of the  $C_2$  allele was zero? Would the population ever be able to adapt to a harsh drought? Explain how genetic diversity is important to adaptation.
4. Is the following statement true or false? Explain. “The population had the genetic diversity to adapt, but could not adapt because the environmental change occurred too abruptly.”
- \*5. (Advanced) Explore the model by modifying the trait size needed for survival (cells C4–E4), initial allele frequencies, and the environmental conditions experienced in years 1–5. Provide an interesting observation in terms of adaptation as a result of your exploration.

## LITERATURE CITED

- Clayton, G. A. and A. Robertson. 1957. An experimental check on quantitative genetical theory. II. Long-term effects of selection. *Journal of Genetics* 55: 152–170.
- Grant, B. R. and P. R. Grant. 1993. Evolution of Darwin’s finches caused by a rare climatic event. *Proceedings of the Royal Society of London (B)* 251:111–117.